



UvA-DARE (Digital Academic Repository)

Epistemic modelling and protocol dynamics

Wang, Y.

Publication date
2010

[Link to publication](#)

Citation for published version (APA):

Wang, Y. (2010). *Epistemic modelling and protocol dynamics*. [Thesis, fully internal, Universiteit van Amsterdam]. Institute for Logic, Language and Computation.

General rights

It is not permitted to download or to forward/distribute the text or part of it without the consent of the author(s) and/or copyright holder(s), other than for strictly personal, individual use, unless the work is under an open content license (like Creative Commons).

Disclaimer/Complaints regulations

If you believe that digital publication of certain material infringes any of your rights or (privacy) interests, please let the Library know, stating your reasons. In case of a legitimate complaint, the Library will make the material inaccessible and/or remove it from the website. Please Ask the Library: <https://uba.uva.nl/en/contact>, or a letter to: Library of the University of Amsterdam, Secretariat, P.O. Box 19185, 1000 GD Amsterdam, The Netherlands. You will be contacted as soon as possible.

Appendix A

Alloy Code for Russian Cards Problem (3.3.1)

```
module RCP
//Alloy program for finding a deterministic protocol solution
//to Russian Cards Problem (3.3.1)
//y.wang@cwi.nl

sig Cards {
}

sig Hands {
  content: set Cards
}

sig Pa {
  member: set Hands
}

fact {all h: Hands | #h.content = 3 }
//every hand contains 3 cards

fact {all p: Pa | #p.member > 1 }
//Any announcement contains at least 2 hands

fact {all h: Hands | some p:Pa | h in p.member }
//Every hand appears at some announcement (executability)

fact {no h: Hands, g: Hands | h != g && h.content = g.content }
//No two hands are the same

fact {no p: Pa, q: Pa |p != q && # p.member & q.member > 0 }
//No two announcements share a hand (for determinism)
```

```
fact {no p: Pa | some h: p.member, g: p.member |
h != g && # h.content & g.content > 1 }
//In order to let B know:
//any two hands in one announcement share at most 1 card

fact {no p: Pa| some c: Cards, d: Cards | all h: p.member|
(not c in h.content) => d in h.content }
//In order to let C stay ignorant: if we fix one card in an announcement
//then the hands that do not contain this card don't have a card in common

pred RCP { }

run RCP for exactly 7 Cards , exactly 35 Hands, 7 Pa
//Given 7 cards there are 35 3-hand. At most 7 announcements.
```